
HET MODELSPOOR-CONTROL-SYSTEEM  
 
Door Carel van den Bergh 
 
 
In het navolgende verhaal zal ik proberen om het MODELSPOOR-CONTROL-
SYSTEEM uit te leggen aan de hand van een voorbeeld. Het MODELSPOOR-
CONTROL-SYSTEEM zal ik in het vervolg gewoon MCS noemen. 
 
Er zijn diverse manieren om een modelspoorbaan te bedienen. Ook krijg ik regelmatig de 
vraag of er een bediening mogelijk is per computer. 
 
Wel nu, met het MCS is het mogelijk om met een kleine interface (de schakeling tussen 
de baan en de computer) bovenstaande vraag te realiseren. 
 
Het MCS bevat een aantal elementen die, in de volgende hoofdstukken uitgebreid aan de 
orde komen, doch die nu in het kort even genoemd worden. 
 
1.Handmatige bediening. 
 
Het moet mogelijk zijn en blijven om een modelspoorbaan met de hand door middel van 
knoppen te bedienen. Ongeacht wat het merk ook moge zijn of worden. 
 
2. Interface. 
 
Daar niet alle computers gelijk zijn, moet het toch mogelijk zijn om er zoveel mogelijk 
afzonderlijk aan te kunnen sluiten. Het MCS maakt gebruik van de parallelle printer 
aansluiting van de computer. Veel computers zijn hiermede uitgerust. 
 
3. Programma. 
 
Meestal is dit het grootste probleem. Het MCS-programma is in MSX-basic geschreven. 
Voor de meeste computer gebruikers zal het geen al te grote problemen opleveren om 
hier en daar de computer-afhankelijke instructies op te sporen en dan te wijzigen. 
 
4. Opmerkingen. 
 
In dit hoofdstuk worden aanpassingen of wijzigingen voor reeds bestaande model 
spoorbanen behandeld. Bv. een baan, met zo'n tiental wissels en signalen, die met het 
door de fabriek uitgebrachte bedieningssysteem al is uitgerust.Ook hier is het MCS 
mogelijk of een gedeelte hiervan. 
 
Het hiervoor beschrevene is toegepast op de E.M.V. demonstratie baan en op mijn eigen 
modelbaan. 
 



 
Hoofdstuk l. Handmatige bediening. 
 
Als handmatige bediening is gekozen voor het NX-systeem. NX staat voor eNtrance en 
eXit. Het Engels voor binnenkomst en uitgaand. 
 
Op een bedieningstafel is het (stations-)sporenplan getekend. Bij elk spoor zijn 
drukknoppen aangebracht. Een knop als vertrekpunt en een knop als eindpunt. Zo is het 
mogelijk om van een spoor te vertrekken naar een ander spoor doormiddel van het 
indrukken van de knoppen. Hieronder het voorbeeld van de E.M.V. demonstratiebaan, 
volgens het NX systeem bestuurd. 
 
 

 
 
Een voorbeeld: 
Men wil een trein van spoor 's1' naar spoor 's4' laten rijden. Dan drukt men op de knop  
behorende bij 's1' en op de knop behorende bij 's4'. Het MCS zorgt ervoor dat wissel 'w3' 
rechtdoor en wissel 'w2' afbuigend gezet worden.Ook wordt het signaal behorende bij 
spoor 's1' op groen (of geel) gezet. De trein mag nu vertrekken. Als nu echter op spoor 
's4' nog een trein zou staan of er wagons afgekoppeld zouden zijn, is dit spoor echter nog 
"bezet". In dit laatste geval zal het MCS ervoor zorgen dat de treinbeweging en het wissel 
zetten niet kan plaatsvinden. Om dit te bereiken is een blok-bezet-melding een noodzaak. 
Men kan dit op diverse manieren bereiken, zoals het toepassen van reed-relais,kontakt-
rails of electronisch. Dit laatste wordt nu op de E.M.V demobaan toegepast. 
 
Daar de treinbeweging langzaam is, zal men redelijk lange tijd op de knoppen moeten 
drukken. Daarom is hier gekozen voor een geheugen, die dit onthoudt. Zo'n geheugen is 
opgebouwd uit een relais en een transistor. Wanneer nu op de juiste knoppen gedrukt 
wordt, wordt het relais bekrachtigd, komt in en houdt zichzelf middels een kontakt vast. 



Hier onder staat het schema van zo'n geheugen-schakeling. 
 
 

 
 
Fig. l  Geheugen-schakeling met een relais en een transistor. Voor het gemak genoemd 
FF-unit.  
 
Op 'S' wordt de combinatie van de knoppen aangesloten, punt 'L' wordt aan de massa 
gelegd, punt 'Vr' wordt aangesloten op de voeding die de relais's van spanning voorziet. 
Op 'R' wordt de blok-bezet-melding aangesloten. Dus voor de eenvoud 'S' aan massa zegt 
relais "in", 'R' aan massa zegt relais "uit". De kontakten 'A' t/m 'I' kunnen gebruikt 
worden om wissels aan te sturen, signalen te bedienen en om de trein te laten rijden (door 
bv. de rails rijspanning te geven). Kontakt 'K' blijft hier vrij. 
 
In ons voorbeeld zijn ook knoppen genoemd, waarop gedrukt kan worden om een 
treintraject te verkrijgen. De verbindingen tussen deze knoppen zijn zeer eenvoudig. Hier 
onder het voorbeeld van de sporen 's1','s2' en 's3' naar 's4'. 
 



 
 
Fig. 2 Bedienings-unit. Voor het gemak genoemd BD-unit.  
 
Met behulp van bovenstaand schema is het al mogelijk om drie treintrajecten te bedienen. 
Voor de E.M.V. demobaan is het aantal treintrajecten tien. Er zijn acht FF-units en acht 
BD-units gebouwd. Zie bijlage I voor een voorbeeld schema. Op de punten SI t/m S3 kan 
de interface worden aangesloten, om zodoende computersturing mogelijk te maken. 



Hoofdstuk 2. Interface. 
 
De interface van het MCS is opgebouwd uit een aantal standaard LOCMOS 
componenten. Het kan ook met andere componenten gebouwd worden. Om een goede 
communicatie met de computer te verkrijgen wordt het signaal van de computer-
printeruitgang eerst "gelatched". Dit wil zeggen onthouden op een commando. Dit 
commando wordt strobe genoemd. Het strobe-signaal staat op punt l van de printer-
uitgang en wordt toevoerd op punt 9 van het "latch" IC (hier de HEF40174). De andere 
signalen van de printer-uitgang worden naar het "latch" IC toegevoerd volgens het 
bijgaand schema. Deze andere signalen worden ook wel "data" genoemd. In dit geval 
bestaat de"data" uit acht "bits". Om nu de "data" uit elkaar te halen, is gebruik gemaakt 
van 3 andere IC's. Een HEF4556, welke dient om "data-bits" 4 en 5 uit te selecteren. De 
HEF4556 stuurt een geselecteerde HEF4051 aan, en deze op zijn beurt selecteert ook 
weer "data-bits" en wel uit de nummers l t/m 3.Daar de geselecteerde signalen van de 
HEF4051 te zwak zijn om verder te gebruiken worden zij middels een transistor 
versterkt. 
 
Hieronder het schema van de interface. 

 
Aan de +5 Volt worden aangesloten:  

a. HEF 40174    punt 1,16  
b. HEF 4556     punt 13,14,15,16  
c. HEF 4051     punt 3,16 

Aan de min worden aangesloten:  
a. HEF 40174    punt 8  
b. HEF 4556     punt 1,8  
c. HEF 4051     punt 8 



Hoofdstuk 3. Programma. 
 
Als we nu het programma laden in onze computer en dan 'run' geven, zal op het scherm 
het volgende verschijnen. 
 

 
Dit gebeurt door het programma gedeelte van regel 10 tot 380. In regel 330 wordt 
gevraagd om uw keuze in te toetsen. De ingetoetste toets wordt in regel 380 ingelezen 
(Zie bijlage A). 
 
Stel, we toetsen op de 'I', dan zal het nu volgende beeld op het scherm verschijnen. Dit 
gebeurt door het programma gedeelte van regel 470 tot en met 510 (Zie bijlage B). 
 

 
Op regel 490 staat GOSUB 960. Hier begint de display routine l, 
welke loopt tot en met 1090. Deze routine tekent op het scherm de 
gele strepen alsmede ook de cijfers 9, 4, 5, 7 en 8. 
Op regel 500 staat GOSUB 1100. Hier begint de display routine 2, 
welke loopt tot en met 1230. Deze routine tekent op het scherm de 
groene strepen alsmede ook de cijfers l, 2, 3 en 6. 
(Zie bijlage C voor de regels). 



 
Na verloop van een seconde verandert de "0" achter "TIJD" in een "l". Weer een seconde 
later zal de "l" in een "2" veranderen en dan "3" en zo verder. Dit gebeuren is een 
meelopende klok en wordt verzorgd door de regels 2370 tot en met 2430 (Zie bijlage E). 
Welke worden aangeroepen in de routine die "van ? " verzorgt, door GOSUB 2370. Deze 
"van ?" routine begint op regel 1750 en loopt tot en met 1870 (Zie bijlage D) en wordt 
aangeroepen op regel 510 door GOSUB 1750. Als we nu op bv 'l' drukken verschijnt 
achter "van ? " de ingedrukte "l" en onder "van ?" de regel "naar ?". Deze laatste routine 
wordt op diverse regels aangeroepen, doch in ons geval door regel 530 en begint op regel 
1880 en loopt tot en met regel 2000. Ook in deze routine wordt weer de klok routine 
aangeroepen door GOSUB 2370. Als we nu op bv '4' drukken verschijnt de "4" achter 
"naar ?" en ook de volgende regel "KODE : 8 TIJD : 3" . Als we nu het programma even 
doorlopen zijn we aangekomen via de regels 520, 530, 540, 550, 740, 750, 760, 770, 780, 
790 op regel 800. Hier staat GOSUB 2320 (Zie bijlage E). Hierop staat PRINT 
#4,CHR$(P(I) + 64);. Dit wil zeggen print het karakter, bepaalt door P(I) samen met 64, 
naar uitgang 4, in ons geval de printer uitgang (welke is vastgelegd in regel 110 (Zie 
bijlage A). Op 2330 wordt een geluids-signaal, op 2340 een wachtlus aangeroepen. Op 
2350 wordt een karakter naar de printer gestuurd en wel CHR$(64) oftewel een "spatie". 
Regel 2360 geeft het einde van de uitgeef routine aan en zijn we terug op regel 810. Na 
wat controlles keert het programma weer terug naar regel 510 en begint opnieuw het "van 
?" en "naar ?" . 
De hierboven genoemde controles bestaan uit het controleren van het aantal combinaties 
(het maximum in dit programma is 100) en of de juiste combinaties wel gemaakt zijn. Is 
een combinatie niet mogelijk , dan verschijnt ook "NIET MOGELIJK" op het scherm. 
Om nu te stoppen met "invoeren" moet de volgende combinatie gemaakt worden "van ? 
l" "naar ? l".Het programma controleert ook deze combinatie, doch geeft geen "NIET 
MOGELIJK" , maar "EINDE" en keert weer terug naar het "MC S-PROGRAMMA 
MENU" , zodat het volgende scherm weer verschijnt. 

 



 
Stel, we toetsen op de 'H', dan zal eerst de volgende vraag gesteld worden : "Hoeveel 
maal ?" . Stel we typen 'l' in,dan zal nu het volgende beeld op het scherm verschijnen. Dit 
gebeurt door het programma gedeelte van regel 1470 tot en met 1740 (Zie bijlage D). 
 

 
Het aantal maal wordt bewaard in "H" op regel 1510. Op de regels 1550, 1560 en 1570 
wordt de opbouw van het scherm aangeroepen. Op regel 1580 wordt via GOSUB 2370 de 
tijd aangeroepen. De volgende regels in het programma kontroleren het maximum aantal , 
de tijd en de kode. Als nu de tijd gelijk is aan de tijd behorende bij de tijd in de 
geheugenplaats "W(I)" , dan zal via de regels 1620 en 1630 weer de printer_routine via 
GOSUB 2320 worden aangeroepen . Dit blijft zich herhalen tot dat een "kode 20" wordt 
gevonden.Dan zal op het scherm "EINDE NR x" verschijnen. In ons geval is de "x" dus 
een "l". Als "x" een "l" is , dan keert het programma terug naar het "MCS-
PROGRAMMA MENU" , zodat het volgende scherm weer verschijnt. 

 
OPMERKING :Bij de herhaling is het mogelijk om dit af te breken, en wel door de 
funktie toets Fl (Zie het bovende staande menu). Ook is het mogelijk om ,wanneer blijkt 
dat een trein beweging te lang op zich laat wachten, dit te corrigeren en wel door de 
funktie toets F5 (Zie het bovende staande menu). 



 
De routine, die funktietoets Fl aanroept, begint op regel 2440. (Zie bijlage E). Hier wordt 
gewoon de "I" op 100 gezet, waardoor het programma afbreekt (Zie regel 1590 op bijlage 
D). De routine, die funktietoets F5 aanroept, begint op regel 2460 en loopt tot en met 
regel 2480 (Zie bijlage E). Op regel 2470 wordt het tijdgeheugen gelijk gemaakt aan de 
lopende tijd. Zodoende wordt deze treinbeweging ook uitgevoerd op dit tijdstip. 
 
Als nu het "MCS-PROGRAMMA MENU" verschenen is na be-eindiging van het 
herhalen, hebben we weer een keuze. De treinen rijden in een volgorde en op gezette 
tijden. Dit "spoorboekje" kunnen we nu onthouden en later opnieuw invoeren of de 
computer laten opslaan. Dit laatste is gemakkelijker. Om dit laatste te doen toetsen we nu 
op 'N'. Al spoedig stelt het programma nu de vraag "Programma nummer" en we toetsen 
nu een 'A' in. Dan verschijnt op het beeld "Programma nummer A" en begint de floppy-
drive te draaien, zodat ons "spoorboekje" op floppy gezet wordt. Deze routine begint op 
regel 2010  (Zie bijlage E).  Is ons "spoorboekje" opgenomen dan verschijnt er op het 
beeld "Opname klaar" en kan weer een keuze gemaakt worden volgens het "MCS-
PROGRAMMA MENU". Stel we zijn klaar en zetten de computer uit. De volgende dag 
willen we demonsteren en zetten de computer weer aan, laden het programma en geven 
'run'. Ook nu zal het "MCS-PROGRAMMA MENU" verschijnen. Nu echter toetsen we 
'V' in. Al spoedig stelt het programma nu de vraag "Programma nummer" en we toetsen 
een 'A' in. Dan verschijnt op het beeld "Programma nummer A" en begint de floppy-drive 
te draaien, zodat ons "spoorboekje" van floppy gehaald wordt. Deze routine begint op 
regel 2150 (Zie bijlage E). Is ons "spoorboekje" ingenomen dan verschijnt er op het beeld 
"Inname klaar". Vervolgens springt het programma naar de 'herhaal' routine. Dit springen 
staat op regel 2280  (Zie bijlage E).De volgende vraag wordt nu gesteld : "Hoeveel maal 
?" . Stel we typen 'l' in, dan zal ook nu het volgende beeld op het scherm verschijnen en 
ons gisteren opgenomen "spoorboekje" wordt uitgevoerd. 

 



 
 
Stel we willen een "spoorboekje" maken zonder dat de treinen gaan rijden. Wat nu te 
doen ? Dit is ook mogelijk. We koppelen als eerste de interface los van de baan. Een 
steker is hier het gemakkelijkst en kiezen volgens het "MCS-PROGRAMMA MENU" 
invoeren. We voeren volgens het vooraf-gaand beschrevene een "spoorboekje" in. Op het 
eind voeren we "van l naar l" in en het "MCS-PROGRAMMA MENU" verschijnt weer 
op het scherm. Vervolgens toetsen we op 'S'. Met deze routine wordt de tijd, waarop de 
treinbewegingen zouden moeten uitgevoerd, op 30 seconden gezet. Het gevolg is dat in 
de "herhaling" elke 30 seconde een nieuwe treinbeweging wordt uitgevoerd en het zo 
makkelijk na te gaan is of dit een goed "spoorboekje zal gaan worden. Is dit 
"spoorboekje" goed, dan kan het worden opgenomen op floppy en later uitgevoerd 
worden. Bij "herhaling" is het mogelijk op de juiste tijd te verkrijgen door het gebruiken 
van funktietoets F5 als men echt wil gaan rijden. 
Ook dit "spoorboekje" kan weer opgenomen worden om later te gebruiken. 
 



Hoofdstuk 4. Opmerkingen. 
 
Als men een bestaand bedieningssysteem heeft, is het MCS of een gedeelte hiervan 
mogelijk. Er zijn diverse methoden mogelijk. Enkele hiervan zullen toegelicht worden. 
Het MCS maakt gebruik van Blok-detectors als bezet melding van een blok. Het gebruik 
van deze Blok-detectors is niet mogelijk (bv. Marklin) of niet meer in een bestaande baan 
in te bouwen. Men kan dan gebruik maken van de kontakt rails, die het gebruikte merk in 
zijn assortiment heeft. Deze kontakt rails kan worden doorverbonden met de reset ingang 
van de FF-unit. Men moet dan wel op de goede verbindingen letten van de voedingen 
en/of transformatoren. Men kan ook gebruik maken van reed-relais om de reset ingang 
aan te sturen. 
Een tweede methode is om geen gebruik te maken van FF-units doch om via een relais-
unit een of meerdere wissels en/of signalen aan te sturen. Bij deze methode fungeert 
alleen de computer als NX bedienings systeem. Dus geen hand NX systeem. 
Een derde methode is om de wissels en/of signalen direct aan te sturen vanuit de 
interface. Men moet hierbij wel rekening houden, dat de versterktrap in de interface nooit 
direct kan aansturen. Men moet hier een zwaardere versterktrap gebruiken. Nog een 
nadeel hierbij is het feit dat de voeding van wissels en/of signalen een gelijkspannings 
voeding moet zijn. 
 
In het verhaal zijn verwijzingen naar bijlagen. 
 
Op bijlage A is een 'listing' van het gebruikte programma afgedrukt. Op de bijlagen F tot 
en met M staan elektrische schema's van de gebruikte schakelingen, zoals van een 
knoppen_paneel, f lip_f lop_unit, blok_detector. Een voorbeeld van de onderlinge 
verbindingen van units is weergegeven op bijlage I. Op bijlage K nog een keer de 
interface en op bijlage M tot slot het schema van het EMV-DEMO-BAAN 
bedieningspaneel . 
Mochten er ondanks dit verhaal nog vragen zijn, dat kan men kontakt opnemen met de 
schrijver. 
Carel van den Bergh 



Bijlage A 
 
10 MAXFILES=5 
20 DIM P(100):DIM W(100) 
30 ON INTERVAL=50 GOSUB 2310 
40 INTERVAL ON 
50 ON KEY GOSUB 2440,,,,2460 
60 KEY(1) ON 
70 KEY(5) ON 
80 SCREEN 2 
90 COLOR 15,1,4 
100 OPEN "GRP:" AS #1 
110 OPEN "LPT:" FOR OUTPUT AS #4 
120 CLS 
130 DRAW "BM50,13" 
140 PRINT #1,"EM" 
150 DRAW "BM55,21" 
160 PRINT #1,"V" 
170 DRAW "BM100,10" 
180 PRINT #1,"Eindhovense" 
190 DRAW "BM100,20" 
200 PRINT #1,"Modelbouw" 
210 DRAW "BM100,30" 
220 PRINT #1,"Vereniging" 
230 DRAW "BM10,60" 
240 PRINT #1,"  MCS-"; 
250 PRINT #1,"PROGRAMMA MENU" 
260 PRINT #1,"" 
270 PRINT #1,"  Invoeren      = I" 
280 PRINT #1,"  Herhaling     = H" 
290 PRINT #1,"  Van floppy    = V" 
300 PRINT #1,"  Naar floppy   = N" 
310 PRINT #1,"  Set tijd op 30 = S" 
320 PRINT #1,"" 
330 PRINT #1,"  toets nu keuze in" 
340 PRINT #1,"" 
350 PRINT #l,"" 
360 PRINT #1,"  F1= Afbreken" 
370 PRINT #1,"  F5= Tijd-instelling" 
380 A$=INKEY$ 
390 IF A$="" THEN GOTO 380 
400 IF A$="I" OR A$="i" THEN GOTO 470 
410 IF A$="V" OR A$="v" THEN GOTO 2150 
420 IF A$="N" OR A$="n" THEN GOTO 2010 
430 IF A$="H" OR A$="h" THEN GOTO 1470 
440 IF A$="S" OR A$="s" THEN GOTO 2490 
450 PRINT #1,"  FOUTE INVOER" 
460 GOTO 120 
470 CLS 
480 LET I=0:LET T=0 
490 GOSUB 960 
500 GOSUB 1100 
510 GOSUB 1750 
520 ON V GOTO 530,570,600,840,630,680,840,710,840 
530 GOSUB 1880 
540 IF N=1 THEN K=20:GOTO 740 



550 IF N=4 THEN K=8:GOTO 740 
560 GOTO 840 
570 GOSUB 1880 
580 IF N=4 THEN K=7:GOTO 740 
590 GOTO 840 
600 GOSUB 1880 
610 IF N=4 THEN K=6:GOTO 740 
620 GOTO 840 
630 GOSUB 1880 
640 IF N=1 THEN K=4:GOTO 740 
650 IF N=2 THEN K=3:GOTO 740 
660 IF N=6 THEN K=2:GOTO 740 
670 GOTO 840 
680 GOSUB 1880 
690 IF N=7 THEN K=1:GOTO 740 
700 GOTO 840 
710 GOSUB 1880 
720 IF N=3 THEN K=5:GOTO 740 
730 GOTO 840 
740 LINE (50,150)-(250,180),1,BF 
750 DRAW "BM50,170" 
760 P(I)=K:W(I)=T 
770 PRINT #1,"KODE :";P(I); 
780 PRINT #1,"    TIJD ";W(I) 
790 IF P(I)=20 THEN GOTO 900 
800 GOSUB 2320 
810 IF I=100 THEN GOTO 900 
820 I=I+1 
830 GOTO 870 
840 LINE (50,150)-(250,180),1,BF 
850 DRAW "BM50,170" 
860 PRINT #1,"NIET MOGELIJK" 
870 GOSUB 2290 
880 LINE (50,150)-(250,180),1,BF 
890 GOTO 510 
900 DRAW "BM50,180" 
910 PRINT #1,"EINDE" 
920 FOR J=1 TO 5 
930 GOSUB 2290 
940 NEXT J 
950 GOTO 120 
960 REM DISPLAY ROUTINE 1 
970 DRAW "BM4,18":PRINT #1,"9" 
980 LINE (10,20)-(70,22),10,BF 
990 LINE (45,10)-(70,12),10,BF 
1000 DRAW "BM4,38":PRINT #1,"4" 
1010 LINE (10,40)-(70,42),10,BF 
1020 DRAW "BM4,48":PRINT #1,"5" 
1030 LINE (10,50)-(70,52),10,BF 
1040 LINE (45,50)-(48,12),10,BF 
1050 LINE (180,40)-(240,42),10,BF 
1060 DRAW "BM244,48":PRINT #1,"7" 
1070 LINE (180,50)-(220,52),10,BF 
1080 DRAW "BM244,38":PRINT #1,"8" 
1090 RETURN 
1100 REM DISPLAY ROUTINE 2 
1110 LINE (75,10)-(125,12),2,BF 



1120 DRAW "BM128,8":PRINT #1,"!" 
1130 LINE (135,10)-(175,12),2,BF 
1140 LINE (75,20)-(125,22),2,BF 
1150 DRAW "BM128,18":PRINT #1,"2" 
1160 LINE (135,20)-(175,22),2,BF 
1170 LINE (75,40)-(125,42),2,BF 
1180 DRAW "BM128,38":PRINT #1,"3" 
1190 LINE (135,40)-(175,42),2,BF 
1200 LINE (75,50)-(125,52),2,BF 
1210 DRAW "BM128,48":PRINT #1,"6" 
1220 LINE (135,50)-(175,52),2,BF 
1230 RETURN 
1240 REM DISPLAY ROUTINE 3 
1250 IF I=101 THEN RETURN 
1260 IF P(I)=20 THEN RETURN 
1270 RESTORE 
1280 LET L=2*P(I)-1 
1290 FOR J=l TO L 
1300 READ V$ 
1310 NEXT J 
1320 READ N$ 
1330 LINE (50,150)-(250,190),1,BF 
1340 DRAW "BM50,150" 
1350 PRINT #1,"van :" 
1360 DRAW "BM108,150" 
1370 PRINT #1,V$ 
1380 DRAW "BM50,160" 
1390 PRINT #l,"Naar :" 
1400 DRAW "BM108,160" 
1410 PRINT #1,N$ 
1420 DRAW "BM50,170" 
1430 PRINT #1,"tijd :" 
1440 DRAW "BM100,170" 
1450 PRINT #1,W(I) 
1460 RETURN 
1470 REM HERHALING 
1480 PRINT #1,"  Hoeveel maal ?" 
1490 A$=INKEY$ 
1500 IF A$="" THEN GOTO 1490 
1510 H=VAL(A$) 
1520 PRINT #1,H 
1530 CLS 
1540 LET I=0:LET T=0 
1550 GOSUB 960 
1560 GOSUB 1100 
1570 GOSUB 1240 
1580 GOSUB 2370 
1590 IF I=100 THEN GOTO 1660 
1600 IF W(I) <> T THEN GOTO 1580 
1610 IF P(I)=20 THEN GOTO 1660 
1620 GOSUB 2370 
1630 GOSUB 2320 
1640 I=I+1 
1650 GOTO 1570 
1660 LINE (50,180)-(250,190),1,BF 
1670 DRAW "BM50,180" 
1680 PRINT #1,"EINDE Nr. ";H 



1690 H=H-1 
1700 IF H > O THEN GOTO 1540 
1710 FOR J=1 TO 5 
1720 GOSUB 2290 
1730 NEXT J 
1740 GOTO 120 
1750 REM INPUT ROUTINE 1 
1760 DRAW "BM50,300" 
1770 PRINT #1,"van  ? "; 
1780 GOSUB 2370 
1790 DRAW "BM100,150" 
1800 A$=INKEY$ 
1810 IF A$="" THEN GOTO 1780 
1820 V=ASC(A$) 
1830 IF V>48 AND V<58 THEN GOTO 1850 
1840 GOTO 1800 
1850 V=V-48 
1860 PRINT #1,A$ 
1870 RETURN 
1880 REM INPUT ROUTINE 2 
1840 DRAW "BM50,160" 
1900 PRINT #1,"naar ? "; 
1910 GOSUB 2370 
1920 DRAW "BM100,160" 
1930 A$=INKEY$ 
1940 IF A$="" THEN GOTO 1910 
1950 N=ASC(A$) 
1960 IF N>48 AND N<58 THEN GOTO 1980 
1970 GOTO 1930 
1980 N=N-48 
1990 PRINT #1,A$ 
2000 RETURN 
2010 REM OPNAME OP FLOPPY 
2020 PRINT #1,"  Programma nummer "; 
2030 C$=INKEY$ 
2040 IF C$="" THEN GOTO 2030 
2050 PRINT #1,C$ 
2060 LET C$="A:PRG"+C$ 
2070 OPEN C$ FOR OUTPUT AS #2 
2080 FOR I=0 TO 99 
2090 PRINT #2,P(I),W(I) 
2100 NEXT I 
2110 CLOSE #2 
2120 PRINT #1,"  Opname klaar " 
2130 GOSUB 2290 
2140 GOTO 120 
2150 REM WEERGAVE VAN FLOPPY 
2160 PRINT #1,"  Programma nummer "; 
2170 C$=INKEY$ 
2180 IF C$="" THEN GOTO 2160 
2190 PRINT #1,C$ 
2200 LET C$="A:PRG"+C$ 
2210 OPEN C$ FOR OUTPUT AS #3 
2220 FOR I=0 TO 99 
2230 INPUT #3,P(I),W(I) 
2240 NEXT I 
2250 CLOSE #3 



2260 PRINT #1,"  Inname klaar" 
2270 GOSUB 2290 
2280 GOTO 1470 
2290 FOR U=0 TO 250:NEXT U 
2300 RETURN 
2310 T=T+1:RETURN 
2320 PRINT #4,CHR$(P(I)+64); 
2330 BEEP:BEEP:BEEP:BEEP:BEEP 
2340 GOSUB 2290 
2350 PRINT #4,CHR$(64); 
2360 RETURN 
2370 REM DISPLAY TIJD 
2380 IF T=S THEN GOTO 2430 
2390 S=T 
2400 LINE (100,120)-(150,130),1,BF 
2410 DRAW "BM50,120" 
2420 PRINT #1," TIJD ";T 
2430 RETURN 
2440 LET I=100 
2450 RETURN 
2460 REM TIJDS CORRECTIE 
2470 LET W(I)=T 
2480 RETURN 
2490 REM TIJD SETTING OP 30 
2500 FOR I=0 TO 100 
2510 IF P(I)=20 THEN GOTO 120 
2520 LET W(I)=30*(I+1) 
2530 NEXT I 
2540 GOTO 120 
2550 DATA 6,7,5,6,5,2,5,1 
2570 DATA 8,3,3,4,2,4,1,4 

